Mat 1. 2-timersprøve den 5. december 2016.

JE 04.12.16

Opgave 1

```
> restart;with(LinearAlgebra):
```

Et inhomogent lineært ligningssystem bestående at tre ligninger med fire ubekendte x_1, x_2, x_3 og x_4 har totalmatricen $\mathbf{T} = [\mathbf{A}|\ \mathbf{b}]$ givet ved

```
> T:=a-><<1,0,0>|<0,a,0>|<1,-2,1-a^2>|<3,-4,0>|<0,2,1-a>>:
>  T(a);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & a & -2 & -4 & 2 \\ 0 & 0 & -a^2+1 & 0 & 1-a \end{bmatrix}$$

hvor $a \in \mathbb{R}$.

Spørgsmål 1

For $a = -1$ er totalmatricen

```
> T(-1);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & -1 & -2 & -4 & 2 \\ 0 & 0 & 0 & 0 & 2 \end{bmatrix}$$

Ligningssystemet har ingen løsninger, da $\rho(\mathbf{A}(-1)) = 2 < 3 = \rho(\mathbf{T}(-1))$.
Eller. Ligningen svarende til den sidste række i $\mathbf{T}$ har ingen løsninger. Derfor har ligningssystemet ingen løsninger.
Med Maple fås

```
> LinearSolve(T(-1));
Error, (in LinearAlgebra:-LinearSolve) inconsistent system
```

For $a = 0$ er totalmatricen

```
> T(0);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & 0 & -2 & -4 & 2 \\ 0 & 0 & 1 & 0 & 1 \end{bmatrix}$$

som har trappeformen

```
> trap('T(0)')=ReducedRowEchelonForm(T(0));
```

$$trap(T(0)) = \begin{bmatrix} 1 & 0 & 0 & 0 & 2 \\ 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

Da koefficientmatricen og totalmatricen begge har rangen $\rho = 3$ og da $n - \rho = 4 - 3 = 1$ er der uendelig mange løsninger karakteriseret ved én fri parameter.

Sættes $x_2 = t$ aflæses umiddelbart, at

$(x_1, x_2, x_3, x_4) = (2, 0, 1, -1) + t(0, 1, 0, 0)$, $t \in \mathbb{R}$.

Med Maple fås

```
> x=LinearSolve(T(0),free=t);
```

$$x = \begin{bmatrix} 2 \\ t_2 \\ 1 \\ -1 \end{bmatrix}$$

For $a = 1$ er totalmatricen

```
> T(1);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & 1 & -2 & -4 & 2 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

som er på trappeform.

Da koefficientmatricen og totalmatricen begge har rangen $\rho = 2$ og da $n - \rho = 4 - 2 = 2$ er der uendelig mange løsninger karakteriseret ved to frie parametre.

Sættes $x_3 = t_1$ og $x_4 = t_2$ aflæses umiddelbart, at

$(x_1, x_2, x_3, x_4) = (0, 2, 0, 0) + t_1(-1, 2, 1, 0) + t_2(-3, 4, 0, 1)$, $t_1, t_2 \in \mathbb{R}$.

Med Maple fås

```
> x=LinearSolve(T(1),free=t);
```

$$x = \begin{bmatrix} -t_3 - 3\,t_4 \\ 2 + 2\,t_3 + 4\,t_4 \\ t_3 \\ t_4 \end{bmatrix}$$

Spørgsmål 2

Koefficientmatricen er

```
> A:=SubMatrix(T(a),1..3,1..4);
```

$$A := \begin{bmatrix} 1 & 0 & 1 & 3 \\ 0 & a & -2 & -4 \\ 0 & 0 & -a^2 + 1 & 0 \end{bmatrix}$$

og højresiden er

```
> b:=SubMatrix(T(a),1..3,5..5);
```

$$b := \begin{bmatrix} 0 \\ 2 \\ 1 - a \end{bmatrix}$$

$\mathbf{x}_0 = (x_1, x_2, x_3, x_4) = (1, 0, -1, 0)$ er en løsning $\Leftrightarrow$ $\mathbf{Ax}_0 = \mathbf{b}$.

```
> x0:=<1,0,-1,0>;
```

$$x0 := \begin{bmatrix} 1 \\ 0 \\ -1 \\ 0 \end{bmatrix}$$

```
> A.x0=b;
```

$$\begin{bmatrix} 0 \\ 2 \\ a^2 - 1 \end{bmatrix} = \begin{bmatrix} 0 \\ 2 \\ 1 - a \end{bmatrix}$$

Heraf ses, at $\mathbf{Ax}_0 = \mathbf{b} \Leftrightarrow a^2 - 1 = 1 - a \Leftrightarrow a^2 + a - 2 = 0 \Leftrightarrow a = 1$ eller $a = -2$.
Altså $(x_1, x_2, x_3, x_4) = (1, 0, -1, 0)$ er en løsning til ligningssystemet $\Leftrightarrow$ $a = 1$ eller $a = -2$.

Opgave 2

```
> restart;with(LinearAlgebra):
```

I $\mathbb{R}^3$ er der givet vektorerne $\mathbf{v}1 = (-1, 1, 0)$, $\mathbf{v}2 = (1, -2, 1)$, $\mathbf{v}3 = (1, -4, 0)$ og $\mathbf{u} = 2\,\mathbf{v}1 - \mathbf{v}2$.

Spørgsmål 1

Koordinatmatricen for vektorerne $\mathbf{v}1$, $\mathbf{v}2$ og $\mathbf{v}3$ med hensyn til standardbasis e er

```
> eV:=<<-1,1,0>|<1,-2,1>|<1,-4,0>>;
```

$$eV := \begin{bmatrix} -1 & 1 & 1 \\ 1 & -2 & -4 \\ 0 & 1 & 0 \end{bmatrix}$$

Da

```
> rho('eV')=Rank(eV);
```

$$\rho(eV) = 3$$

er $\mathbf{v}1$, $\mathbf{v}2$ og $\mathbf{v}3$ tre lineært uafhængige vektorer i $\mathbb{R}^3$ og dermed er $v = (\mathbf{v}1, \mathbf{v}2, \mathbf{v}3)$ en basis for $\mathbb{R}^3$.
Basisskiftematricen der skifter fra v-koordinater til standard e-koordinater er

```
> eMv:=eV;
```

$$eMv := \begin{bmatrix} -1 & 1 & 1 \\ 1 & -2 & -4 \\ 0 & 1 & 0 \end{bmatrix}$$

og basisskiftematricen der skifter fra standard e-koordinater til v-koordinate er

```
> vMe:=eMv^(-1);
```

$$vMe := \begin{bmatrix} -\frac{4}{3} & -\frac{1}{3} & \frac{2}{3} \\ 0 & 0 & 1 \\ -\frac{1}{3} & -\frac{1}{3} & -\frac{1}{3} \end{bmatrix}$$

Spørgsmål 2

Koordinatvektoren for $\mathbf{u} = 2\mathbf{v}1 - \mathbf{v}2$ med hensyn til basis v er

```
> vu:=<2,-1,0>;
```

$$vu := \begin{bmatrix} 2 \\ -1 \\ 0 \end{bmatrix}$$

Koordinatvektoren for $\mathbf{u}$ med hensyn til standardbasis e er

```
> eu:=eMv.vu;
```

$$eu := \begin{bmatrix} -3 \\ 4 \\ -1 \end{bmatrix}$$

Dette kunne også fås ved

```
> eu:=2*<-1,1,0>-<1,-2,1>;
```

$$eu := \begin{bmatrix} -3 \\ 4 \\ -1 \end{bmatrix}$$

$f : \mathbb{R}^3 \to \mathbb{R}^3$ er lineær og der gælder, at $f(\mathbf{v}1) = 3\mathbf{v}1$, $f(\mathbf{v}2) = 3\mathbf{v}2$ og $f(\mathbf{v}3) = \mathbf{0}$.

Spørgsmål 3

Da de tre vektorer er egentlige vektorer, så følger det af det givne, at $\mathbf{v}1$ og $\mathbf{v}2$ er egenvektorer for f hørende til egenværdien 3 og at $\mathbf{v}3$ er en egenvektor for f hørende til egenværdien 0.
$v = (\mathbf{v}1, \mathbf{v}2, \mathbf{v}3)$ er således en basis for $\mathbb{R}^3$ bestående af egenvektorer for f .

Spørgsmål 4

Da $f(\mathbf{v}1) = 3\mathbf{v}1 = 3\ \mathbf{v}1 + 0\ \mathbf{v}2 + 0\ \mathbf{v}3$, $f(\mathbf{v}2) = 3\mathbf{v}2 = 0\ \mathbf{v}1 + 3\ \mathbf{v}2 + 0\ \mathbf{v}3$ og $f(\mathbf{v}3) = \mathbf{0} = 0\ \mathbf{v}1 + 0\ \mathbf{v}2 + 0\ \mathbf{v}3$ er

```
> vFv:=DiagonalMatrix(<3,3,0>);
```

$$vFv := \begin{bmatrix} 3 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> eFe:=eMv.vFv.vMe;
```

$$eFe := \begin{bmatrix} 4 & 1 & 1 \\ -4 & -1 & -4 \\ 0 & 0 & 3 \end{bmatrix}$$

Spørgsmål 5

Da både **v**1 og **v**2 tilhører egenvektorrummet E_3 (de udspænder E_3), så vil også linearkombinationen $\mathbf{u} = 2\,\mathbf{v}1 - \mathbf{v}2$ tilhøre E_3.

u er altså en egenvektor for f hørende til egenværdien 3.

Kontrol

```
> eFe.eu=3*eu;
```

$$\begin{bmatrix} -9 \\ 12 \\ -3 \end{bmatrix} = \begin{bmatrix} -9 \\ 12 \\ -3 \end{bmatrix}$$

Opgave 3

```
> restart;with(LinearAlgebra):
```

$f : C^{\infty}(\mathbb{R}) \rightarrow C^{\infty}(\mathbb{R})$ lineær og givet ved

```
> f:=x->diff(x,t)-t*x:
> 'f(x(t))'=f(x(t));
```

$$f(x(t)) = \frac{d}{dt}\, x(t) - t\, x(t)$$

Spørgsmål 1

```
> f(x(t))=0;
```

$$\frac{d}{dt}\, x(t) - t\, x(t) = 0$$

Da

```
> Int(-t,t)=int(-t,t);
```

$$\int (-t)\, dt = -\frac{1}{2}\, t^2$$

er samtlige løsninger

```
> x(t)=c*exp(1/2*t^2);
```

$$x(t) = c\, e^{\frac{1}{2}\, t^2}$$

hvor $t \in \mathbb{R}$ og $c \in \mathbb{R}$.

Direkte med Maple fås

```
> dsolve(f(x(t))=0,x(t));
```

$$x(t) = \_C1\, e^{\frac{1}{2} t^2}$$

$P_2(\mathbb{R})$ med monomiebasis $m = (1, t, t^2)$ og $P_3(\mathbb{R})$ med monomiebasis $m = (1, t, t^2, t^3)$.

Vi begrænser nu f til at være en afbildning af $P_2(\mathbb{R})$ ind i $C^\infty(\mathbb{R})$.

Spørgsmål 2

Billederne af basisvektorerne er

```
> f(1);
```

$$-t$$

```
> f(t);
```

$$-t^2 + 1$$

```
> f(t^2);
```

$$-t^3 + 2\,t$$

Da $f(1)$, $f(t)$ og $f(t^2)$ alle tilhører $P_3(\mathbb{R})$, så er billedrummet $f(P_2(\mathbb{R}))$, som er udspændt af disse tre billedvektorer, et underrum af $P_3(\mathbb{R})$.

Inspireret af spørgsmål 2, lader vi til sidst f være en afbildning af $P_2(\mathbb{R})$ ind i $P_3(\mathbb{R})$.

Spørgsmål 3

Da $f(1) = -t$, $f(t) = 1 - t^2$ og $f(t^2) = 2t - t^3$ ifølge spørgsmål 2, så aflæser vi, at

```
> mFm:=<<0,-1,0,0>|<1,0,-1,0>|<0,2,0,-1>>;
```

$$mFm := \begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 2 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

$P(t) \in \ker(f) \Leftrightarrow f(P(t)) = 0\mathit{1} + 0t + 0t^2 + 0t^3 \Leftrightarrow \mathrm{m}\mathbf{F}\mathrm{m}\ \mathrm{mP}(t) = \mathbf{0} \in \mathbb{R}^4$.

```
> LinearSolve(<mFm|<0,0,0,0>>);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Dvs $\ker(f)$ består kun af nulpolynomiet $0\mathit{1} + 0\,t + 0\,t^2$ i $P_2(\mathbb{R})$.

Af dimensionssætningen følger så, at $\dim(f(P_2(\mathbb{R}))) = \dim(P_2(\mathbb{R})) - \dim(\ker(f)) = 3 - 0 = 3$.

(Eller $\dim(f(P_2(\mathbb{R}))) = \rho(\mathrm{m}\mathbf{F}\mathrm{m}) = 3$)

Sammenligning med spørgsmål 1: Forskellen skyldes, at den eneste af funktionerne $c\,\exp(\frac{1}{2}\,t^2)$, der ligger i $P_2(\mathbb{R})$, som vi afbilder fra, er den, der svarende til $c = 0$.

Opgave 4

```
> restart;with(LinearAlgebra):with(plots):
```

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \mathbf{A} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, \ t \in \mathbb{R}.$$

hvor **A** er en reel 2×2-matrix. Det oplyses endvidere, at **A** har egenvektoren

```
> v1:=<1+I,2>;
```

$$v1 := \begin{bmatrix} 1 + \mathrm{I} \\ 2 \end{bmatrix}$$

hørende til egenværdien

```
> lambda1:=-1+2*I;
```

$$\lambda 1 := -1 + 2\,\mathrm{I}$$

Spørgsmål 1

Da **A** er en reel 2×2-matrix, så er den anden egenværdi for **A** den konjugerede af λ1

```
> lambda2:=conjugate(lambda1);
```

$$\lambda 2 := -1 - 2\,\mathrm{I}$$

og en tilhørende egenvektor for **A** er den konjugerede af **v**1

```
> v2:=evalc(conjugate(v1));
```

$$v2 := \begin{bmatrix} 1 - \mathrm{I} \\ 2 \end{bmatrix}$$

Da **v**1 og **v**2 er to lineært uafhængige egenvektorer for **A** er den fuldstændige komplekse løsning til differentialligningssystemet

$$\mathbf{x}(t) = c1\, e^{(-1+2i)t}\,\mathbf{v}1 + c2\, e^{(-1-2i)t}\,\mathbf{v}2 \ , \ t \in \mathbb{R} \ , \ c1, c2 \in \mathbb{C}.$$

Skrevet ud i Maple

```
> x:=unapply(c1*exp(lambda1*t)*v1+c2*exp(lambda2*t)*v2,t):
> 'x(t)'=x(t);
```

$$x(t) = \begin{bmatrix} (1+\mathrm{I})\, c1\, \mathrm{e}^{(-1+2\mathrm{I})\,t} + (1-\mathrm{I})\, c2\, \mathrm{e}^{(-1-2\mathrm{I})\,t} \\ 2\, c1\, \mathrm{e}^{(-1+2\mathrm{I})\,t} + 2\, c2\, \mathrm{e}^{(-1-2\mathrm{I})\,t} \end{bmatrix}$$

Det bemærkes, at **x**(t) er reel, hvis og kun hvis den er lig sin konjugerede hvilket er opfyldt, hvis og kun hvis $c1$ og $c2$ er indbyrdes konjugerede.

Spørgsmål 2

Af figuren aflæses, at **x**(0) = $(x_1(0), x_2(0))$ = (-1, 4).

Til bestemmelse af konstanterne *c1* og *c2* haves det lineære ligningssystem

```
> x(0)=<-1,4>;
```

$$\begin{bmatrix} (1+\mathrm{I})\, c1 + (1-\mathrm{I})\, c2 \\ 2\, c1 + 2\, c2 \end{bmatrix} = \begin{bmatrix} -1 \\ 4 \end{bmatrix}$$

med tilhørende totalmatrix

```
> T:=<<1+I,2>|<1-I,2>|<-1,4>>;
```

$$T := \begin{bmatrix} 1+\mathrm{I} & 1-\mathrm{I} & -1 \\ 2 & 2 & 4 \end{bmatrix}$$

som har trappeformen

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 1+\frac{3}{2}\,\mathrm{I} \\ 0 & 1 & 1-\frac{3}{2}\,\mathrm{I} \end{bmatrix}$$

Heraf aflæses, at

```
> c1:=1+3/2*I;
```

$$c1 := 1 + \frac{3}{2}\,\mathrm{I}$$

og

```
> c2:=1-3/2*I;
```

$$c2 := 1 - \frac{3}{2}\,\mathrm{I}$$

c1 og *c2* er indbyrdes konjugerede som forventet. Jævnfør bemærkningen under spørgsmål 1.
Den søgte løsning er da

```
> 'x(t)'=simplify(evalc(x(t)));
```

$$x(t) = \begin{bmatrix} -\mathrm{e}^{-t}\left(\cos(2\,t) + 5\sin(2\,t)\right) \\ 2\,\mathrm{e}^{-t}\left(2\cos(2\,t) - 3\sin(2\,t)\right) \end{bmatrix}$$

eller skrevet ud

```
> 'x[1](t)'=simplify(evalc(x(t)[1]));
```

$$x_1(t) = -\mathrm{e}^{-t}\left(\cos(2\,t) + 5\sin(2\,t)\right)$$

```
> 'x[2](t)'=simplify(evalc(x(t)[2]));
```

$$x_2(t) = 2\,\mathrm{e}^{-t}\left(2\cos(2\,t) - 3\sin(2\,t)\right)$$

hvor $t \in \mathbb{R}$.

```
> p1:=plot(x(t)[1],t=0..6,color=red):
> p2:=plot(x(t)[2],t=0..6,color=blue):
> display(p1,p2,scaling=constrained,view=-3..5);
```



5
4
3
2
1
0
−1
−2
−3
1
2
3
4
5
6
t